

Paper 259-2012

## Graphing Made Easy with SG Procedures

Lora D. Delwiche, University of California, Davis, CA

Susan J. Slaughter, Avocet Solutions, Davis, CA

### ABSTRACT

New with SAS® 9.2, ODS Graphics introduces a whole new way of generating graphs using SAS. With just a few lines of code, you can create a wide variety of high-quality graphs. This paper covers the three basic Statistical Graphics (SG) procedures—SGPLOT, SGPANEL and SGSCATTER. SGPLOT produces single-celled graphs. SGPANEL produces multi-celled graphs that share common axes. SGSCATTER produces multi-celled graphs that may use different axes. This paper shows how to use each of these procedures to produce different types of graphs, how to send your graphs to different ODS destinations, how to access individual graphs, and how to specify properties of graphs, such as format, name, height and width.

### INTRODUCTION

When ODS Graphics was originally conceived, the goal was to enable statistical procedures to produce sophisticated graphs tailored to each specific statistical analysis, and to integrate those graphs with the destinations and styles of the Output Delivery System. Starting with SAS 9.2, over 60 statistical procedures have the ability to produce graphs using ODS Graphics. A fortuitous side effect of all this graphical power has been the creation of a set of procedures for creating stand-alone graphs (graphs that are not embedded in the output of a statistical procedure). These procedures all have names that start with the letters SG.

This paper compares the three procedures that create graphics—SGPLOT, SGPANEL, and SGSCATTER. SGPLOT creates one or more graphs and overlays them on a single set of axes. SGPANEL is similar to SGPLOT except that SGPANEL produces a panel of graphs based on one or more classification variables. In SGPANEL all the graphs must use the same axes. SGSCATTER also produces a panel of graphs, but those graphs may use different axes. SGPLOT and SGPANEL share similar statements, but SGSCATTER uses a separate set of statements.

The SGRENDER and SGDESIGN procedures are different. They do not create graphs; instead they render graphs that have been previously defined. SGRENDER renders graphs from custom ODS graph templates that you create using the TEMPLATE procedure. SGDESIGN renders graphs from SGD files that you create using the ODS Graphics Designer, a point-and-click application. SGRENDER and SGDESIGN are not covered in this paper.

This paper was written using SAS 9.3, but almost all the features discussed here also work in SAS 9.2. For those features that are new with SAS 9.3, we note the differences in their descriptions.

### ODS GRAPHICS VS. TRADITIONAL SAS/GRAPH

To use ODS Graphics in SAS 9.2 you must have SAS/GRAPH software which is licensed separately from Base SAS. Starting with SAS 9.3, ODS Graphics is part of Base SAS. Some people may wonder whether ODS Graphics replaces traditional SAS/Graph procedures. No doubt, for some people and some applications, it will. But ODS Graphics is not designed to do everything that traditional SAS/Graph does, and does not replace it.

Here are some of the major differences between ODS Graphics and traditional SAS/GRAPH procedures.

#### Traditional SAS/GRAPH


- Graphs are saved in SAS graphics catalogs
- Graphs are viewed in the Graph window
- Can use GOPTIONS statements to control appearance of graphs

#### ODS Graphics

- Graphs are saved in standard image file formats such as PNG and JPEG
- Graphs are viewed in standard viewers such as a web browser for HTML output
- Can use ODS GRAPHICS statements to control appearance of graphs

## VIEWING ODS GRAPHICS

How you view your graphs depends on what version of SAS you are using. Starting with SAS 9.3, HTML is the default ODS destination in the windowing environment. So any graph you produce using SAS 9.3 will appear in the Results Viewer window automatically. When you produce ODS Graphics in the SAS 9.2 windowing environment, for most output destinations a window opens to display your results. However, when you use the LISTING destination, graphs are not automatically displayed. Regardless of its destination, you can always view a graph by double-clicking its graph icon or name in the Results window.



Display 1. Results Window

## EXAMPLES AND DATA

The purpose of this paper is to compare the three major SG procedures—SGPLOT, SGPANEL, and SGSCATTER. The following examples show a small sample of the types of graphs these procedures can produce, and are designed to illustrate default behaviors. As you read this, keep in mind that there are many other types of graphs (especially for SGPLOT and SGPANEL) and many options. The tables at the end of this paper list all the basic statements and many of the options. For examples using options and optional statements for changing the appearance of plots, see Slaughter & Delwiche (2010).

The examples in this paper use data about the 196 countries that participated in the 2008 Olympics.

## SGPLOT PROCEDURE

SGPLOT produces many types of graphs. In fact, this one procedure produces 18 different types of graphs and more types continue to be added with new releases of SAS. SGPLOT creates one or more graphs and overlays them on a single set of axes. There are four axes in a set: left, right, top, and bottom.

## BAR CHARTS

Bar charts show the distribution of a categorical variable. This example uses a VBAR statement to create a vertical bar chart of the variable Region. The bars represent the number of countries in each region.

```
PROC SGPLOT DATA = Countries;
  VBAR Region;
  TITLE 'Olympic Countries by '
    'Region';
RUN;
```




Figure 1. VBAR Plot

This bar chart is like the first one except that the bars have been divided into groups using the GROUP= option. The grouping variable is a categorical variable named PopGroup. The GROUP= option can be used with most types of plots in SGPLOT, SGPANEL, and SGSCATTER.

```
PROC SGPLOT DATA = Countries;
  VBAR Region / GROUP = PopGroup;
  TITLE 'Olympic Countries by Region '
    'and Population Group';
RUN;
```




Figure 2. VBAR with GROUP= Option

In the following code, the GROUP= option has been replaced with a RESPONSE= option. The response variable is NumParticipants, the number of participants in the 2008 Olympics from each country. Now each bar represents the total number of participants for a region.

```
PROC SGPLOT DATA = Countries;
  VBAR Region /
    RESPONSE = NumParticipants;
  TITLE 'Olympic Participants by '
    'Region';
RUN;
```




Figure 3. VBAR with RESPONSE= Option

## HISTOGRAMS AND DENSITY PLOTS

Histograms show the distribution of a continuous variable. Density plots show standard distributions (either NORMAL or KERNEL) for the data, and are often drawn on top of histograms.

When overlaying plots, the order of the statements determines which plot is drawn on top. The plot resulting from the first statement will be on the bottom, followed by the second, and so on. Care must be taken to make sure that subsequent plots do not obscure the first.

In this example, the HISTOGRAM statement draws the distribution of the variable, TotalMedals, which is the total number of medals won by each country. The DENSITY statement overlays a density plot on top of the histogram. The default density plot is the normal distribution.

```

PROC SGPLOT DATA = Countries;
  HISTOGRAM TotalMedals;
  DENSITY TotalMedals;
  TITLE 'Total Medals for '
        'Olympic Countries';
RUN;

```




Figure 4. HISTOGRAM with DENSITY Overlaid

## SCATTER PLOTS

Scatter plots show the relative distribution of two continuous variables on Cartesian coordinates. To create a scatter plot, use a SCATTER statement. This example produces a scatter plot of the number of participants per country versus the total number of medals won.

```

*Scatter plot using SGPLOT;
PROC SGPLOT DATA = Countries;
  SCATTER X = NumParticipants
          Y = TotalMedals;
  TITLE 'Number of Participants '
        'by Total Medals Won '
        'for Each Country';
RUN;

```




Figure 5. SCATTER Plot

## SGPANEL PROCEDURE

The SGPANEL procedure produces nearly all the same types of graphs as SGPLOT, but instead of displaying only one plot per image, SGPANEL displays a panel of plots in a single image. A separate plot is produced for each level of the variable you specify in the PANELBY statement.

The syntax for SGPANEL is almost identical to SGPLOT, so it is easy to convert SGPLOT code to SGPANEL by making just a couple of changes to your code. You simply replace the SGPLOT keyword with SGPANEL, and add a PANELBY statement. The PANELBY statement must appear before any statements that create plots. This example takes the preceding SGPLOT example and converts it to an SGPANEL. The PANELBY statement produces a separate plot for each value of Region.

```

*Scatter plots using SGPANEL;
PROC SGPANEL DATA = Countries;
  PANELBY Region;
  SCATTER X = NumParticipants
          Y = TotalMedals;
  TITLE 'Number of Participants '
        'by Total Medals Won '
        'for Each Country';
RUN;

```




Figure 6. SGPANEL with SCATTER Plot

The variable Region used in the PANELBY statement happens to have an even number of values so it creates a nice rectangular panel. If your PANELBY variable has an uneven number of values, then you may end up with empty cells in your panel. You can use the ROWS= and COLUMNS= options to change the shape of your panel. You may also want to use the NOVARNAM= option to remove the name of the PANELBY variable from the graph, and the SPACING= option to separate the cells making them more distinct.

```

*SGPANEL with options;
PROC SGPANEL DATA = Countries;
  PANELBY Region / SPACING = 5
  NOVARNAM= ROWS=6;
  SCATTER X = NumParticipants
          Y = TotalMedals;
  TITLE 'Number of Participants '
        'by Total Medals Won';
RUN;

```




Figure 7. SGPANEL with Options on PANELBY Statement

## SGSCATTER PROCEDURE

At first glance, the SGSCATTER procedure appears similar to SGPANEL because both procedures produce multi-celled graphs. However, while SGPLOT and SGPANEL share the same statements for producing plots, SGSCATTER uses a completely different (and much smaller) set of statements. In fact, SGSCATTER has only three plot statements—COMPARE, MATRIX, and PLOT. SGSCATTER cannot create bar charts or box plots, or many of the other types of plots created by SGPLOT and SGPANEL. True to its name, SGSCATTER is designed primarily for scatter plots. SGSCATTER also offers fewer options for controlling the layout and appearance of your graphs. However, SGSCATTER does have one special ability. SGSCATTER can combine plots that use different axes.

The COMPARE statement creates a paneled graph that uses common axes for each row and column of cells. The result is similar to graphs created by SGPANEL except that in SGSCATTER each cell represents a different combination of variables whereas in SGPANEL the data represent the same variables (grouped according to the value of the PANELBY variable).

```
PROC SGSCATTER DATA = Countries;
  COMPARE Y = TotalMedals
  X = (Population NumParticipants);
  TITLE 'Total Medals by Number of
  'Participants and Population '
  'for Each Country';
RUN;
```




Figure 8. SGSCATTER with COMPARE

The MATRIX statement creates a scatter plot matrix. Each variable appears on both the vertical and horizontal axes so that all the variables are crossed with each other. By default, the diagonal cells (which cross a variable with itself) will be empty. The DIAGONAL= option is often used to add a histogram and density curve in these empty cells.

```
PROC SGSCATTER DATA = Countries;
  MATRIX Population NumParticipants
  TotalMedals /
  DIAGONAL = (HISTOGRAM NORMAL);
  TITLE 'Scatter Plot Matrix for '
  'Population, Number of '
  'Participants, and Total Medals';
RUN;
```




Figure 9. SGSCATTER with MATRIX

The PLOT statement creates a paneled graph where the axes for each cell are completely independent. In this example, NumParticipants is crossed with TotalMedals, and Gold (the number of gold medals won) is crossed with Silver (the number of silver medals won).

```
PROC SGSCATTER DATA = Countries;
  PLOT NumParticipants * TotalMedals
      Gold * Silver;
  TITLE 'Side-by-Side Plots of '
        'Number of Participants by Total '
        'Medals and Gold by Silver';
RUN;
```




Figure 10. SGSCATTER with PLOT

## CHANGING THE ODS STYLE

The following code creates a bar chart using the default ODS style template and destination. In SAS 9.3, the default destination is HTML and the default STYLE is HTMLBLUE. In SAS 9.2 the default destination is LISTING. In the LISTING destination, tabular output is rendered as plain text so styles apply only to graphical output. The default style for the LISTING destination is named LISTING.

```
* Bar chart with default style;
PROC SGPLOT DATA = Countries;
  VBAR Region / GROUP = PopGroup;
  TITLE 'Olympic Countries by Region '
        'and Population Group';
RUN;
```




Figure 11. Default Style for LISTING Destination

You can use the STYLE= option in an ODS destination statement to specify a style for your output including graphs. The following ODS LISTING statement changes the style to JOURNAL.

```

* Change ODS style template;
ODS LISTING STYLE = JOURNAL;
PROC SGPLOT DATA = Countries;
  VBAR Region / GROUP = PopGroup;
  TITLE 'Olympic Countries by Region '
        'and Population Group';
RUN;

```




Figure 12. JOURNAL Style

You can use the same style templates for graphs as you do for tabular output. However, some styles are better suited for statistical graphs than others. The following table lists some of the styles that are recommended for graphical results.

<u>Desired Output</u>	<u>Style Template</u>	<u>Default for Destination</u>
Color	ANALYSIS DEFAULT HTMLBLUE LISTING PRINTER RTF STATISTICAL	HTML (in SAS 9.2) HTML (in SAS 9.3) LISTING (graphs only) PRINTER, PDF, PS RTF
Gray scale	JOURNAL	
Black and white	JOURNAL2	

Every destination has a default style associated with it. This means that if you change the destination for a graph, its appearance may change too.


## CHANGING THE ODS DESTINATION

You can send ODS Graphics output to any ODS destination, and you do it in the same way that you would for tabular output, using ODS statements for that destination. These statements send a bar chart to the PDF destination.

```

* Send graph to PDF
destination;
ODS PDF FILE =
'c:\MyPDFFiles\BarChart.pdf';
PROC SGPLOT DATA = Countries;
  VBAR Region /
  GROUP = PopGroup;
  TITLE 'Olympic Countries by '
        'Region and Population '
        'Group';
RUN;
ODS PDF CLOSE;

```



Display 2. PDF Results Displayed in Results Viewer

## SAVING ODS GRAPHICS OUTPUT AND ACCESSING INDIVIDUAL GRAPHS

For most destinations (including RTF and PDF), graphs are integrated with tabular output into a single file. For these destinations, you can use the FILE= option to tell SAS where to save your output. This statement would create PDF output and save it in file named Report.pdf in a folder named MyPDFFiles.

```
ODS PDF FILE = 'c:\MyPDFFiles\Report.pdf';
```

However, when you send output to the LISTING or HTML destinations, graphs are saved in individual files separate from tabular output. By default, in SAS 9.3, these files are saved in the same folder as your SAS WORK library. In SAS 9.2 these files are saved in your current SAS working directory. (In the SAS windowing environment, the path of the current SAS working directory appears in the lower-right corner of the SAS window.) For the LISTING and HTML destinations, you can use the GPATH= option to tell SAS where to save individual graphs. This statement would save all graphs sent to the HTML destination in a folder named MyGraphs.

```
ODS HTML GPATH = 'c:\MyGraphs';
```

For the HTML destination, these individual graph files are linked to the tabular output so that they appear integrated when viewed in a Web browser. Therefore, when you send graphical output to the HTML destination, you must be careful to save those files in directories where they can be found by anyone who displays the HTML file in a Web browser.

## SPECIFYING PROPERTIES OF IMAGES

Using the ODS GRAPHICS statement, you can control many aspects of your graphics files including the image format, name, height, and width. LISTING is the best destination for this since it offers the most image formats, and saves images in separate files. Here are a few of the options you can specify in the ODS GRAPHICS statement:

<u>Option</u>	<u>Description</u>	<u>Example</u>
IMAGEFMT=	specifies image format	IMAGEFMT = JPEG
IMAGENAME=	specifies base filename for image	IMAGENAME = 'NewGraph'
HEIGHT=	specifies height	HEIGHT = 4IN
WIDTH=	specifies width	WIDTH = 8CM
RESET	resets any preceding options to default values	RESET

Image formats supported for the LISTING destination include PNG (the default), BMP, GIF, JPEG, JPG, PDF, PS, TIFF, and many others.

The default name for an image file is the name of its ODS output object. (You can use ODS TRACE statements to find the name of any ODS output object.) SAS will append numerals to the end of the image name. For example, if you specify an image name of Olympic, then the files will be named Olympic, Olympic1, Olympic2, Olympic3, and so on. If you rerun your code, SAS will, by default, continue counting so that your graphics files will not be overwritten. If you want to start over each time you run your code, then specify the RESET option before the IMAGENAME= option.

In most cases, the default size for graphs is 640 pixels wide by 480 pixels high. If you specify only one dimension (width but not height, or vice versa), then SAS will adjust the other dimension to maintain a default aspect ratio of 4/3. You can specify width and height in these units: CM, IN, MM, PT, or PX.

The following code creates a graph in JPEG format that is 2 high inches by 3 inches wide. The file will be named Final.jpeg and will be saved in a folder named MyGraphs. In such a small image, a legend can take an inordinate amount of space. You can eliminate the legend by adding the NOAUTOLEGEND option to the PROC statement.

```
* Change image properties;
ODS GRAPHICS / RESET IMAGENAME = 'Final'
  IMAGEFMT =JPEG HEIGHT = 2in WIDTH = 3in;
ODS LISTING GPATH = 'c:\MyGraphs' ;

PROC SGPLOT DATA = Countries NOAUTOLEGEND;
  HISTOGRAM TotalMedals;
  DENSITY TotalMedals;
  TITLE 'Total Medals for Olympic '
    'Countries';
RUN;
```




Figure 13. JPEG Image with HEIGHT and WIDTH Options Applied

## SG PROCEDURE SYNTAX TABLES

The following six tables summarize the statements and options for the SG procedures.

	<u>Procedures</u>	<u>Description</u>
<b>Table 1</b>	SGPLOT, SGPANEL	Compatibility and syntax of plot statements and options. This table shows which statements and options can be used together.
<b>Table 2</b>	SGPLOT, SGPANEL	Individual plot statements with syntax and selected options. The options listed in this table are in addition to the options listed in Table 1.
<b>Table 3</b>	SGPLOT	Optional statements and selected options to control axes, or add reference lines and insets.
<b>Table 4</b>	SGPANEL	Required and optional statements and selected options to control panels, axes, or add reference lines.
<b>Table 5</b>	SGSCATTER	Individual plot statements with syntax and selected options.
<b>Table 6</b>	SGPLOT, SGPANEL, SGSCATTER	Attribute options for plot statements. Note that it is also possible to use ODS styles to control these attributes, or to change them using the ODS Graphics Editor.




Even with all the options listed in these tables, this is just a sample. Each plot statement has many possible options—we have listed only a few of the most useful. For a complete list of available options, see the SAS Help and Documentation.

**Table 1.** Compatibility and syntax of SGPLOT and SGPANEL procedure statements and selected options.

	SCATTER	SERIES	STEP	NEEDLE	VECTOR	BUBBLE	BAND	HIGHLOW	REG	LOESS	PBSPLINE	ELLIPSE <sup>1</sup>	HBOX/VBOX	HISTOGRAM	DENSITY	HBAR/VBAR	HLINE/VLINE	DOT
<b>Basic X Y Plots</b>																		
<i>PLOTNAME X=var Y=var / options;</i>																		
SCATTER	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓						
SERIES	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓						
STEP	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓						
NEEDLE	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓						
VECTOR	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓		✓	✓			
<i>BUBBLE X=var Y=var SIZE=var / options;</i>																		
BUBBLE	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓						
<b>Band and Highlow Plots</b>																		
<i>BAND X=var UPPER=var LOWER=var / options; (You can also specify numeric values for upper and lower.)</i>																		
BAND	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓						
<i>HIGHLOW X=var HIGH=var LOW=var / options; (You can also specify numeric values for high and low.)</i>																		
HIGHLOW	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓						
<b>Fit and Confidence Plots</b>																		
<i>PLOTNAME X=var Y=var / options;</i>																		
REG	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓						
LOESS	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓						
PBSPLINE	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓						
ELLIPSE <sup>1</sup>	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓						
<b>Distribution Graphs – Continuous DATA</b>																		
<i>PLOTNAME response-var / options;</i>																		
HBOX or VBOX													✓	✓	✓			
HISTOGRAM					✓										✓			
DENSITY					✓									✓	✓			
<b>Distribution Graphs – Categorical DATA</b>																		
<i>PLOTNAME category-var / options;</i>																		
HBAR or VBAR																✓	✓	✓
HLINE or VLINE																✓	✓	✓
DOT																✓	✓	✓
<b>Selected Options in Plot Statements</b>																		
GROUP <i>/GROUP=var;</i>	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓				✓	✓	✓
TRANSPARENCY <i>/TRANSPARENCY=value;</i>	✓	✓	✓	✓	✓	✓	✓	✓				✓	✓	✓	✓	✓	✓	✓
MARKERS <i>/MARKERS;</i>		✓	✓	✓													✓	
LEGENDLABEL <i>/LEGENDLABEL= 'text-string';</i>	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓	✓
FILLATTRS <i>/FILLATTRS= (attribute=val);</i>						✓	✓					✓		✓		✓		
LINEATTRS <i>/LINEATTRS= (attribute=val);</i>		✓	✓	✓	✓	✓	✓		✓	✓	✓	✓			✓		✓	
MARKERATTRS <i>/MARKERATTRS= (attribute=val);</i>	✓	✓	✓	✓					✓	✓	✓						✓	✓





<sup>1</sup> The ELLIPSE statement is not available for SGPANEL.

Table 2. SGPLOT and SGPANEL plot statements and selected options.

	SYNTAX	SELECTED OPTIONS	
<b>SCATTER</b> 	SCATTER X=var Y=var/options;	DATALABEL=var	Displays a label for each DATA point
<b>SERIES</b> 	SERIES X=var Y=var/options;	BREAK	Creates a break in the line for each missing value
		CURVELABEL	Labels the series curve using the Y variable label
<b>STEP</b> 	STEP X=var Y=var/options;	BREAK	Creates a break in the line for each missing value
		CURVELABEL	Labels the step curve using the Y variable label
<b>NEEDLE</b> 	NEEDLE X=var Y=var/options;	BASELINE=val	Specifies a numeric value on the Y axis for the baseline
<b>VECTOR</b> <sup>1</sup> 	VECTOR X=var Y=var/options;	XORIGIN=val	Specifies X coordinate for origin, either numeric value or numeric variable.
		YORIGIN=val	Specifies Y coordinate for origin, either numeric value or numeric variable.
<b>BUBBLE</b> <sup>1</sup> 	Bubble X=var Y=var Size=var/options;	FILL NOFILL	Specifies if fill is visible or not
		OUTLINE NOOUTLINE	Specifies if outline is visible or not
<b>BAND</b> 	BAND X=var UPPER=var LOWER=var/options;	FILL NOFILL	Specifies if fill is visible or not
		OUTLINE NOOUTLINE	Specifies if outline is visible or not
<b>HIGHLOW</b> <sup>1</sup> 	HIGHLOW X=var HIGH=var LOW=var/options;	HIGHCAP=val LOWCAP=val TYPE=val	Specifies type of cap to use at end of bar or line  Specifies how to draw plot, with a: BAR or LINE(default)




<sup>1</sup> The VECTOR statement is new with SAS 9.2 Phase 2, the BUBBLE and HIGHLOW statements are new with SAS 9.3.

Table 2 (continued). SGPLOT and SGPANEL plot statements and selected options.

	SYNTAX	SELECTED OPTIONS	
<b>REG</b> 	REG X=var Y=var/options;	ALPHA=val	Specifies confidence level (default 0.05)
		CLI	Displays confidence limits for individual predicted values
		CLM	Displays confidence limits for mean predicted values
<b>LOESS</b> 	LOESS X=var Y=var/options;	ALPHA=val	Specifies confidence level (default 0.05)
		CLM	Displays confidence limits
		INTERPOLATION=	Specifies degree of interpolation: CUBIC (default) or LINEAR
<b>PBSPLINE</b> 	PBSPLINE X=var Y=var/options;	ALPHA=val	Specifies confidence level (default 0.05)
		CLI	Displays confidence limits for individual predicted values
		CLM	Displays confidence limits for mean predicted values
<b>ELLIPSE<sup>1</sup></b> 	ELLIPSE X=var Y=var/options;	ALPHA=val	Specifies confidence level for the ellipse
		TYPE=	Specifies type of ellipse: MEAN or PREDICTED (default)
<b>HBOX/VBOX</b> 	VBOX response-var/options; HBOX response-var/options;	CATEGORY=var	A box plot is created for each value of the category variable
		MISSING	Creates box plot for missing values of category variable
<b>HISTOGRAM</b> 	HISTOGRAM response-var/options;	SHOWBINS <sup>1</sup>	Places tick mark at midpoint of bin
		SCALE=	Specifies scale for vertical axis: PERCENT (default), COUNT or PROPORTION
<b>DENSITY</b> 	DENSITY response-var/options;	SCALE=	Specifies scale for vertical axis: DENSITY (default), PERCENT COUNT or PROPORTION
		TYPE=	Specifies type of density function: NORMAL (default) or KERNEL

<sup>1</sup> The ELLIPSE statement and SHOWBINS option are not available for SGPANEL.

**Table 2** (continued). SGPLOT and SGPANEL plot statements and selected options.




	SYNTAX	SELECTED OPTIONS	
<b>HBAR/VBAR<sup>1</sup></b> 	<code>VBAR category-var/options;</code> <code>HBAR category-var/options;</code>	<code>RESPONSE=var</code> <code>STAT=</code> <code>BARWIDTH=num</code>	Specifies a numeric response variable for plot Specifies statistic for axis of response variable (if specified): MEAN, or SUM (default) Specifies numeric value for width of bars (default 0.8)
<b>HLINE/VLINE</b> 	<code>VLINE category-var/options;</code> <code>HLINE category-var/options;</code>	<code>RESPONSE=var</code> <code>STAT=</code>	Specifies numeric response variable for plot Specifies statistic for axis of response variable (if specified): MEAN, or SUM (default)
<b>DOT</b> 	<code>DOT category-var/options;</code>	<code>RESPONSE=var</code> <code>STAT=</code> <code>LIMITSTAT=</code>	Specifies numeric response variable for plot Specifies statistic for axis of response variable (if specified): MEAN, or SUM (default) Specifies statistic for limit lines (must use STAT=MEAN): CLM (default), STDDEV, or STDERR

<sup>1</sup> To make bar charts of pre-summarized data, use HBARPARM or VBARPARM available starting with SAS 9.3.

**Table 3.** Selected optional statements for SGPLOT with selected options.




	SYNTAX	SELECTED OPTIONS	
<b>REFLINE</b> 	<code>REFLINE value1 value2 .../ options;</code>	<code>AXIS=</code>	Specifies axis for reference line: X, Y (default), X2, or Y2
		<code>LABEL= ( )</code>	Creates labels for reference lines: ( <i>'text1' 'text2' ...</i> )
		<code>LABELLOC=</code>	Specifies placement of label with respect to plot area: INSIDE (default) or OUTSIDE
		<code>LINEATTRS= ( )</code>	Specifies attributes for reference line: ( <i>attribute=value</i> )
<b>XAXIS/YAXIS</b> 	<code>XAXIS options; YAXIS options;</code>	<code>GRID</code>	Creates grid line at each tick on the axis
		<code>LABEL=' text'</code>	Specifies a label for the axis
		<code>TYPE=</code>	Specifies type of axis: DISCRETE, LINEAR, LOG, or TIME
		<code>VALUES= ( )</code>	Specifies values for ticks on the axis: ( <i>num1,num2,...</i> ) or ( <i>num1 TO num2 BY increment</i> )
<b>INSET</b> 	<code>INSET 'text1' 'text2' ... / options;</code>	<code>BORDER</code>	Creates a border around text box
		<code>POSITION=</code>	Specifies position of text box within plot area: BOTTOM, BOTTOMLEFT, BOTTOMRIGHT, LEFT, RIGHT, TOP, TOPLEFT, or TOPRIGHT

**Table 4.** Required and selected optional statements for SGPANEL with selected options.

	SYNTAX	SELECTED OPTIONS	
<b>Required</b>			
<b>PANELBY</b>	<code>PANELBY var1 var2 .../ options;</code>	<code>COLUMNS=</code>	Specifies number of columns
		<code>ROWS=</code>	Specifies number of rows
		<code>LAYOUT=</code>	Determines how panels are laid out: PANEL (default), LATTICE, COLUMNLATTICE, or ROWLATTICE <sup>1</sup>
		<code>SPACING=</code>	Specifies number of pixels between rows and columns (default 0)
		<code>NOVARNAME</code>	Eliminates the variable name in the heading for the cell
<b>Optional</b>			
<b>REFLINE</b>	<code>REFLINE value1 value2 .../ options;</code>	<code>AXIS=</code>	Specifies axis for reference line: X, Y (default), X2, or Y2
		<code>LABEL=( )</code>	Creates labels for reference lines: ('text1' 'text2' ...)
		<code>LINEATTRS=( )</code>	Specifies attributes for reference line: (attribute=value)
<b>COLAXIS/ ROWAXIS</b>	<code>COLAXIS options; ROWAXIS options;</code>	<code>GRID</code>	Creates grid line at each tick on the axis
		<code>LABEL=' text'</code>	Specifies a label for the axis
		<code>TYPE=</code>	Specifies type of axis: DISCRETE, LINEAR, LOG, or TIME
		<code>VALUES=( )</code>	Specifies values for ticks on the axis: (num1,num2,...) or (num1 TO num2 BY increment)

<sup>1</sup> The COLUMNLATTICE and ROWLATTICE options are new with SAS 9.2 Phase 2.

**Table 5.** Plot statements for SGSCATTER with selected options.

	SYNTAX	SELECTED OPTIONS	
<b>COMPARE</b>			
	<code>COMPARE X=var Y=yar/ options;</code>	<code>GROUP=</code>	Specifies classification variable to group values
		<code>ELLIPSE</code>	Adds confidence or prediction ellipse to plot
		<code>JOIN<sup>1</sup></code>	Joins points with straight lines
		<code>REG</code>	Adds a regression fit to plots
		<code>LOESS</code>	Adds loess fit curves to plots
		<code>PBSPLINE</code>	Adds penalized B-spline fit curves to plots
		<code>GRID</code>	Creates grid lines for each tick
		<code>MARKERATTRS=</code>	Specifies color, size, and/or symbol for markers. See table 6.
<b>MATRIX</b>			
	<code>MATRIX var1 var2 .../ options;</code>	<code>GROUP=</code>	Specifies classification variable to group values
		<code>DIAGONAL=( )</code>	Specifies graphs to add to the diagonal cells of plot matrix: HISTOGRAM, KERNEL and/or NORMAL
		<code>ELLIPSE</code>	Adds confidence or prediction ellipse to plot
		<code>MARKERATTRS=</code>	Specifies color, size, and/or symbol for markers. See table 6.
<b>PLOT</b>			
	<code>PLOT (yvar1 yvar2 ...)* (xvar1 xvar2 ...)/ options;</code>	<code>GROUP=</code>	Specifies classification variable to group values
		<code>COLUMNS=</code>	Specifies number of columns
		<code>ROWS=</code>	Specifies number of rows
		<code>UNISCALE=</code>	Specifies which axes should be scaled uniformly for all cells: X, Y, or ALL. (default no scaling)
		<code>ELLIPSE</code>	Adds confidence or prediction ellipse to plot
		<code>JOIN<sup>1</sup></code>	Joins points with straight lines
		<code>REG</code>	Adds a regression fit to plots
		<code>LOESS</code>	Adds loess fit curves to plots
		<code>PBSPLINE</code>	Adds penalized B-spline fit curves to plots
		<code>GRID</code>	Creates grid lines for each tick
		<code>MARKERATTRS=</code>	Specifies color, size, and/or symbol for markers. See table 6.

<sup>1</sup> The JOIN option is new with SAS 9.2 Phase 2.

**Table 6.** The attribute options for plot statements.

	SYNTAX	ATTRIBUTES	
<b>FILLATTRS</b> 	<code>/FILLATTRS=(attribute=value);</code>	<code>COLOR=</code>	Specifies color for fill including: AQUA, BLACK, BLUE, FUCHSIA, GREEN, GRAY, LIME, MAROON, NAVY, OLIVE, PURPLE, RED, SILVER, TEAL, WHITE, and YELLOW
<b>LINEATTRS</b> 	<code>/LINEATTRS=(attribute=value);</code>	<code>COLOR=</code>	Specifies color for fill including: AQUA, BLACK, BLUE, FUCHSIA, GREEN, GRAY, LIME, MAROON, NAVY, OLIVE, PURPLE, RED, SILVER, TEAL, WHITE, and YELLOW
		<code>PATTERN=</code>	Specifies pattern for line including: SOLID, DASH, SHORTDASH, LONGDASH, DOT, DASHDASHDOT, and DASHDOTDOT
		<code>THICKNESS=val</code>	Specifies thickness of line. Value can include units: CM, IN, MM, PCT, PT, or PX (default)
<b>MARKERATTRS</b> 	<code>/MARKERATTRS=(attribute=value);</code>	<code>COLOR=</code>	Specifies color for fill including: AQUA, BLACK, BLUE, FUCHSIA, GREEN, GRAY, LIME, MAROON, NAVY, OLIVE, PURPLE, RED, SILVER, TEAL, WHITE, and YELLOW
		<code>SIZE=val</code>	Specifies size of marker. Value can include units: CM, IN, MM, PCT, PT, or PX (default)
		<code>SYMBOL=</code>	Specifies symbol for marker including: CIRCLE, CIRCLEFILLED, DIAMOND, DIAMONDFILLED, PLUS, SQUARE, SQUAREFILLED, STAR, STARFILLED, TRIANGLE, and TRIANGLEFILLED

## CONCLUSION

ODS Graphics and the SG procedures introduce an exciting new way of producing high-quality graphs using SAS. While the SG procedures don't completely replace traditional SAS/GRAPH procedures, they do offer a wide variety of graphs using simple syntax. Of the three major SG procedures, SGPLOT is the one that people will use most often. Once you have learned SGPLOT, you can easily produce a panel of plots by converting your SGPLOT code to SGPANEL. If you need to create multi-celled plots that use different axes, then you will want to use SGSCATTER. Because ODS Graphics is part of the Output Delivery System, you can use the same styles and destinations that you use for tabular output. Using the ODS GRAPHICS statement, you can control many properties of your graphs.

## REFERENCES

- Central Intelligence Agency. 2007. "The World Factbook." <https://www.cia.gov/library/publications/the-world-factbook/index.html>.
- Delwiche, L. D. Slaughter, S. J. 2008. *The Little SAS Book: A Primer, Fourth Edition*. SAS Institute, Cary, NC.
- Mantange, S. Heath, D. 2011. *Statistical Graphics Procedures by Example: Effective Graphs Using SAS*. SAS Institute, Cary, NC.
- SAS Institute Inc. 2008. "SAS/GRAPH 9.2: Statistical Graphics Procedures Guide." <http://support.sas.com/documentation/onlinedoc/graph/index.html>.
- SAS Institute Inc. 2011. *SAS® 9.3 ODS Graphics: Procedures Guide, Second Edition*. Cary, NC: SAS Institute Inc.
- Slaughter, S. J. & Delwiche, L. D. 2010. Using PROC SGPLOT for Quick High-Quality Graphs. *Proceedings of the SAS Global Forum Conference*, Seattle, WA, SAS Institute, Cary, NC, paper no. 154-2010.

## CONTACT INFORMATION

Lora Delwiche and Susan Slaughter are the authors of *The Little SAS Book: A Primer*, and *The Little SAS Book for Enterprise Guide* which are published by SAS Institute. The authors may be contacted at:

Susan J. Slaughter	<a href="mailto:susan@avocetsolutions.com">susan@avocetsolutions.com</a>
Lora D. Delwiche	<a href="mailto:lldelwiche@ucdavis.edu">lldelwiche@ucdavis.edu</a>

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.